

# 配方施肥建议卡的程序设计

修士忠 张凤霞 柳 莉

(白山市农业技术推广站,白山 134300)

目前我省许多县乃至乡都有了微型电子计算机,为计算机技术应用于农业,奠定了良好的基础。应用微机提出配方施肥建议,处理大量的土壤养分测定数据,既准确又方便。

我们根据配方施肥的养分平衡法原理,利用微机的 BASIC 语言,编制出施肥建议的电算程序。该程序应用范围不受处理样本数的限制,并且直接以施肥建议卡的形式输出,操作十分简便。本程序已在 IBM-PC/XT 微机上运行通过。

## 1 算法概要

本程序根据养分平衡法原理设计,是以土壤养分测定值来计算土壤供肥量的。肥料需要量计算公式为:

$$\text{肥料需要量} = \frac{\text{目标产量} \times \text{作物单位产量养分吸收量} - \text{土壤养分测定值} \times 2.25 \times \text{校正系数}}{\text{肥料养分含量} \times \text{肥料当季利用率}}$$

式中 2.25 是 1 公顷表土量折算后的养分换算系数;土壤养分测定值以 mg/kg 表示;作物单位产量养分吸收量以每公斤产量吸收量计算;目标产量为当地前 3 年作物的平均产量再增加 10%~15%,单位为公斤/公顷;校正系数及肥料当季利用率应根据各地的土壤情况及大量的试验来确定。

## 2 程序设计

### 2.1 变量说明

由键盘输入程序清单。Y——每公顷的目标产量,单位:公斤/公顷;WN,WP,WK——每公斤作物产量吸收的氮、磷、钾养分量;LN,LP,LK——氮、磷、钾化肥的当季利用率;JN,JP,JK——氮、磷、钾土壤养分的校正系数;X——输入的土壤样本数;RN,RP,RK——存放速效氮、磷、钾的土壤测定值;A\$——存放作物名称;B\$——存放户主姓名;C\$——存放地块名称;D\$——存放土壤类型名称。

### 2.2 运行实例

如六道江镇胜利一村农户姜庆发的冷浆型水稻土,土壤养分测定值为:碱解氮 134.95mg/kg、速效磷 14.5mg/kg、速效钾 51mg/kg。

从第 1000 DATA 语句依次输入字符串,户主姓名、地块名称、土壤类型及数字变量碱解氮、速效磷、速效钾的土壤测定值:1000 DATA“姜庆发”,“六道江胜利一村”,“冷浆型水稻土”,134.95,14.5,51。输入完毕后,开始运行(RUN),根据屏幕提示,键入以下项目:作物种类? 水稻↵/Y=? 7500↵/WN,WP,WK=? 0.018,0.01,0.025↵/LN,LP,LK=? 0.3,0.3,0.5↵/JN,JP,JK=? 0.3,0.8,0.4↵/X=? 1↵便可直接打印出施肥建议卡。

## 配方施肥建议卡

(单位:kg/ha)

编号	姓名 姜庆发	地块名称 六道江胜利一村	土壤类型 冷浆型水稻土	作物种类 水稻
目标产量:7500		常年产量:		
土壤测定值: 碱解氮:134.95mg/kg		速效磷:14.5mg/kg	速效钾:51mg/kg	
建议施肥量:				
纯 N,146	纯 P,163	纯 K,283		
硝 铵,429	过 石,1019	硫酸钾,566		
氢 铵,859	三 料,354	氯化钾,472		
尿 素,317	二 铵,354			
注:磷、钾肥做底肥一次投入,氮肥底追施用相结合				

## 2.3 程序清单

5 REM 配方施肥建议卡

10 CLEAR

20 INPUT“作物种类”;A\$

30 INPUT“Y=”;Y

40 INPUT“WN,WP,WK=”;WN,WP,WK

50 INPUT“LN,LP,PK=”;LN,LP,LK

60 INPUT“JN,JP,JK=”;JN,JP,JK

70 INPUT“X=”;X

80 FOR I=1 TO X

90 READ B\$,C\$,D\$

100 READ RN,RP,RK

110 N=INT(Y\*WN-RN\*2.25\*JN)/LN+0.5);IF N&lt;=0 THEN N=0

120 P=INT(Y\*WP-RP\*2.25\*JP)/LP+0.5);IF P&lt;=0 THEN P=0

130 K=INT(Y\*WK-RK\*2.25\*JK)/LK+0.5);IF K&lt;=0 THEN K=0

140 LPRINT TAB(34)“配方施肥建议卡”TAB(60)“(单位:kg/ha)”

150 GOSUB 370

160 LPRINT TAB(1)“编号”TAB(13)“姓名”TAB(28)“地块名称”TAB(51)“土壤类型”TAB(66)“作物种类”

170 LPRINT TAB(13);B\$TAB(28);C\$TAB(51);D\$TAB(66);A\$

180 GOSUB 370

190 LPRINT“目标产量:”;Y TAB(40)“常年产量:”

200 GOSUB 370

210 LPRINT“土壤测定值:”

220 LPRINT“碱解氮:”;RN;“mg/kg”TAB(27)“速效磷:”RP;“mg/kg”TAB(52)“速效钾:”;RK;“mg/kg”

230 GOSUB 370

240 LPRINT“建议施肥量:”

250 LPRINT“纯 N:”;N TAB(27)“纯 P:”;PTAB(52)“纯 K:”;K

260 A=INT(N/0.34+0.5);B=INT(P/0.16+0.5);C=INT(K/0.5+0.5)

270 LPRINT“硝铵:”;A TAB(27)“过石:”;B TAB(52)“硫酸钾:”;C

280 D=INT(N/0.17+0.5);E=INT(P/0.46+0.5);F=INT(K/0.6+0.5)

290 LPRINT“氢铵:”;D TAB(27)“三料:”;E TAB(52)“氯化钾:”;F

300 G=INT(N/0.46+0.5);H=INT(P/0.46+0.5)

310 LPRINT“尿素:”;G TAB(27)“二铵:”;H

320 LPRINT“注:磷、钾肥做底肥一次投入,氮肥底追施用相结合”

330 GOSUB 370

340 LPRINT;LPRINT;LPRINT

350 NEXT I

360 END

370 FOR Z=1 TO 74:LPRINT“-”;NEXT Z

380 RETURN

1000 DATA“姜庆发”,“六道江胜利一村”,“冷浆型水稻土”,134.95,14.5,51

## 参 考 文 献

1 农业部农业局编. 配方施肥. 农业出版社. 1989

2 张苏华. BASIC 语言程序的设计. 山东大学出版社. 1988